
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of ESM test case 10.7.1 which are part of the LTE test suite.
The test case can be demonstrated to run with 1 LTE UE (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Cases:
tc_10_7_1
Test Group:
ESM
ATS Version:
iwd-EUTRA-TVB2009-12_D10wk22
System Simulator used:
Anite Conformance Toolset
UE(s) used:
LG L2000 UE
Verification Status:
PASS
4 Corrections required for test case 10.7.1
4.1 Introduction

This section describes the changes required to make test case 10.7.1 run correctly with a LTE UE. 

4.2 Change 1

	Testcase name
	fl_TC_10_7_1_Body

	Reason for change
	As per the expected sequence

“The trigger in step 1 and the RRC messages in steps 2 to 5 are the same as in the generic procedure in 36.508 clause 6.4.3.1”.
In the current TTCN implementation RRC signalling for steps 2a4 to 2a7 from 36.508 table 6.4.3.1-1 is missing.

	Summary of change
	Added steps to configure the default bearer.

	Source of change
	File : ESM_Testcases.ttcn

	MCC160 Comment
	Accepted in principle. Equivalent to implementation in 10wk26.


Before:
	function fl_TC_10_7_1_Body() runs on EUTRA_PTC

  {

    var  ProcedureTransactionIdentifier v_EpsTi := '00'O;

    var NAS_MSG_Indication_Type v_NasInd;

    // Step 1. Cause the UE to request bearer resource allocation of

    // dedicated EPS bearer associated with first PDN connectivity

    f_UT_RequestMODedicatedBearer (UT,

                                   hex2int(tsc_EpsDedicatedBearerId),

                                   hex2int(tsc_EpsDefaultBearerId),

                                   cs_508_EPS_QoS_Dedicated_1,

                                   cs_508_Tft_Dedicated_1(tsc_EpsDedicatedBearerId));

    // step 2. The UE transmits a SERVICE REQUEST message.

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               cr_EstablishmentCause_MoData,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cr_508_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get())));

    // Step 3. Check: Does the UE transmit a BEARER RESOURCE ALLOCATION REQUEST message? (1,P)

    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_BearerResourceAllocReq(v_EpsTi,

                                                                                    tsc_EpsDefaultBearerId))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");


After:
	function fl_TC_10_7_1_Body() runs on EUTRA_PTC

  {

    var  ProcedureTransactionIdentifier v_EpsTi := '00'O;
    var NAS_MSG_Indication_Type v_NasInd;

    var NasCount_Type v_NasCountUL;                                        

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();
    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_CellInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info ( eutra_CellA);

    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_CellA);
    // Step 1. Cause the UE to request bearer resource allocation of

    // dedicated EPS bearer associated with first PDN connectivity

    f_UT_RequestMODedicatedBearer (UT,

                                   hex2int(tsc_EpsDedicatedBearerId),

                                   hex2int(tsc_EpsDefaultBearerId),

                                   cs_508_EPS_QoS_Dedicated_1,

                                   cs_508_Tft_Dedicated_1(tsc_EpsDedicatedBearerId));

    // step 2. The UE transmits a SERVICE REQUEST message.

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               cr_EstablishmentCause_MoData,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cr_508_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get())));

    // activate security in AS with parameters from AKA performed in preamble

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, 

                                                     v_SecurityParams,

                                                     v_NasCountUL);

    f_EUTRA_Security_Set(v_SecurityParams);
    SRB.send ( cas_SRB1_RrcPdu_REQ ( eutra_CellA,

                                     cs_TimingInfo_Now,

                                     cs_RRCConnectionReconfiguration_SRB2_1AM_DRB_Est ( tsc_RRC_TI_Def,

                                                                                        v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                        v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                        v_AntennaInfo) ) );
    SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_CellA, cr_508_RRCConnectionReconfigurationComplete ( tsc_RRC_TI_Def ) ) );
    // Step 3. Check: Does the UE transmit a BEARER RESOURCE ALLOCATION REQUEST message? (1,P)

    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_BearerResourceAllocReq(v_EpsTi,

                                                                                    tsc_EpsDefaultBearerId))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");
    


4.3 Change 2

	Testcase name
	fl_TC_10_7_1_Body

	Reason for change
	1) At step 3 PTI for BEARER RESOURCE ALLOCATION REQUEST can be any value from 1 to 254.

2) At step 3 BEARER RESOURCE ALLOCATION REQUEST should be received on SRB2 as a result of above change.
3) At step 3 BEARER RESOURCE ALLOCATION REQUEST QOS is of type LV format, hence IEI should be omitted. 
4) At step 4 PTI for ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST should be same as PTI received in BEARER RESOURCE ALLOCATION REQUEST as per the Spec 36.508 from table 4.7.3-3.
5) At step 5 ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT should be received for the dedicated bearer that is activated.

	Summary of change
	1) Initialised v_EpsTi with ’?’.
2) In step 3 replaced car_SRB1_NasPdu_IND with car_SRB2_NasPdu_IND.
3) In step 3 Replaced cr_508_BearerResourceAllocReq with cr_508_BearerResourceAllocReq_Mod.
4) Assigned v_EpsTi with v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.bEARER_RESOURCE_ALLOCATION_REQUEST.procedureTransactionIdentifier before step 4.
5) Replaced tsc_EpsDefaultBearerId with tsc_EpsDedicatedBearerId at step 5.

	Source of change
	File : ESM_Testcases.ttcn, EPS_NAS_Templates.ttcn

	MCC160 Comment
	1) Accepted in principle. Implementation as in 10wk26 used.
2) Accepted in principle. Equivalent to implementation in 10wk26.
3) Accepted in principle. Equivalent to implementation in 10wk26.
4) Accepted.
5) Accepted.


Before:
	function fl_TC_10_7_1_Body() runs on EUTRA_PTC

  {

    var  ProcedureTransactionIdentifier v_EpsTi := '00'O;
// Step 3. Check: Does the UE transmit a BEARER RESOURCE ALLOCATION REQUEST message? (1,P)

    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_BearerResourceAllocReq(v_EpsTi,

                                                                                    tsc_EpsDefaultBearerId))));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");

    // Step 4. The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST message.

    f_EUTRA_TxDedicatedEpsBearerCtxtReq(eutra_CellA,

                                        tsc_DRB2,

                                        tsc_EpsDedicatedBearerId,

                                        v_EpsTi,

                                        cs_LinkedId(hex2bit( tsc_EpsDefaultBearerId)),

                                        cs_508_EPS_QoS_Dedicated_1,

                                        cs_508_Tft_Dedicated_1(tsc_EpsDedicatedBearerId));

    // Step 5. Check: Does the UE transmit an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message? (2,P)

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(tsc_EpsDefaultBearerId))));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");

  } // function fl_TC_10_7_1_Body


After:
	  function fl_TC_10_7_1_Body() runs on EUTRA_PTC

  {

    var  ProcedureTransactionIdentifier v_EpsTi := ?;
    var SRB_COMMON_IND v_ReceivedAsp;
    // Step 3. Check: Does the UE transmit a BEARER RESOURCE ALLOCATION REQUEST message? (1,P)

        SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_BearerResourceAllocReq_Mod(v_EpsTi,

                                                                                    tsc_EpsDefaultBearerId))))-> value v_ReceivedAsp;    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");

    v_EpsTi := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.bEARER_RESOURCE_ALLOCATION_REQUEST.procedureTransactionIdentifier;
    // Step 4. The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST message.

    f_EUTRA_TxDedicatedEpsBearerCtxtReq(eutra_CellA,

                                        tsc_DRB2,

                                        tsc_EpsDedicatedBearerId,

                                        v_EpsTi,

                                        cs_LinkedId(hex2bit( tsc_EpsDefaultBearerId)),

                                        cs_508_EPS_QoS_Dedicated_1,

                                        cs_508_Tft_Dedicated_1(tsc_EpsDedicatedBearerId));

    // Step 5. Check: Does the UE transmit an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message? (2,P)

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(tsc_EpsDedicatedBearerId))));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");

  } // function fl_TC_10_7_1_Body    


New Templates:
	 template EPS_QualityOfService cr_EPS_QoS_Any_Mod :=

{ /* 24.301 cl. 9.9.4.3 */

  iei := omit,

  iel := ?,

  qci                    := ?,

  maxBitRateUl           := *,

  maxBitRateDl           := *,

  guaranteedBitRateUl    := *,

  guaranteedBitRateDl    := *,

  maxBitRateUlExt        := *,

  maxBitRateDlExt        := *,

  guaranteedBitRateUlExt := *,

  guaranteedBitRateDlExt := *

};

template NAS_UL_Message_Type cr_508_BearerResourceAllocReq_Mod(template (value) ProcedureTransactionIdentifier p_ProcTi,

                                                           EPS_BearerIdentity p_LinkedEpsBearerId) :=

{

  bEARER_RESOURCE_ALLOCATION_REQUEST := {

    epsBearerId                    := '0'H,

    protocolDiscriminator          := tsc_PD_ESM,

    procedureTransactionIdentifier := p_ProcTi,

    messageType                    := tsc_MT_BearerResourceAllocationRequest,

    spareHalfOctet                 := tsc_SpareHalfOctet,

    linkedEpsBearerId              := cs_LinkedId(hex2bit(p_LinkedEpsBearerId)),   // of the associated dedicated bearer

    trafficFlowAgg                 := cr_Tft ('001'B),  // Create new TFT

    requiredTrafficFlowQoS         := cr_EPS_QoS_Any_Mod,

    protocolConfigurationOptions   := cr_ProtocolConfigOptionsAny ifpresent

  }

};


4.4 Change 3
	Testcase name
	fl_TFTPort

	Reason for change
	Double quotes missing before Port No in packet filters, for AT commands

	Summary of change
	Added a Double quotes before Port No in the function fl_TFTPort().

	Source of change
	File : UpperTesterFunctions.ttcn

	MCC160 Comment
	Accepted in principle.


Before:
	  function fl_TFTPort (octetstring p_Input) return charstring {

    var integer v_Length := lengthof (p_Input);

    var charstring v_Result := "";

    v_Result := """";

    if (v_Length == 4) {

      v_Result := int2str(oct2int(substr(p_Input, 0, 2)));

      v_Result := v_Result & "." & int2str(oct2int(substr(p_Input, 2, 2)));

    } else {

      v_Result := int2str(oct2int(p_Input)) & "." & int2str(oct2int(p_Input)); // only given one value, so put the same in both

    }

    v_Result := v_Result & """";
    return v_Result;

  }


After: 
	function fl_TFTPort (octetstring p_Input) return charstring {

    var integer v_Length := lengthof (p_Input);

    var charstring v_Result := "";

    v_Result := """";

    if (v_Length == 4) {

      v_Result := int2str(oct2int(substr(p_Input, 0, 2)));

      v_Result := v_Result & "." & int2str(oct2int(substr(p_Input, 2, 2)));

    } else {

      v_Result := int2str(oct2int(p_Input)) & "." & int2str(oct2int(p_Input)); // only given one value, so put the same in both

    }

    v_Result := """" & v_Result & """";
    return v_Result;

  }


5 Execution Log Files
5.1 LG L2000 UE
The LG L2000 UE passed these test cases on the Anite Conformance Toolset solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files tc_10_7_1_LG-log.html:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

6 References

	[1]
	R5s100363: This archive comprises text format execution log file and the TTCN file.

